

Computing Flux Integrals

©2006 Mike May, S.J.- maymk@slu.edu

[>

In considering flux integrals, it seems worthwhile to go through using Maple to set up and evaluate flux integrals over a parameterized surface. We will walk through a step by step procedure, then produce new procedures that do everything in one command. Finally we look at the Maple command that computes the flux integral in a single step. We use the Student[VectorCalculus] package and set the BasisFormat variable so that fields are written as vectors rather than as linear combinations of a basis.

```
> restart:with(Student[VectorCalculus]);
   BasisFormat(false):
[&x, `*`, `+`, `-`, `.`; <, >, <|>, BasisFormat, Binormal, CrossProduct, Curl, Curvature,
  D, Del, Divergence, DotProduct, FlowLine, Flux, GetCoordinates, Gradient, Laplacian,
  LineInt, MapToBasis, Nabla, Norm, Normalize, PathInt, PositionVector,
  PrincipalNormal, RadiusOfCurvature, RootedVector, ScalarPotential, SetCoordinates,
  SpaceCurve, SpaceCurveTutor, SurfaceInt, TNBFrame, TangentVector, Torsion, Vector,
  VectorField, VectorFieldTutor, VectorPotential, diff, evalVF, int, limit, series] (1)
```

A step by step example

First we need to define a vector field that we want to integrate, the parameterized surface we are integrating over, and the limits on the parameters. For the example we will start with the vector field $[0, 0, 1]$. Then the flux integral will be the area of the shadow of the surface on the x-y plane.

```
> vfield := <0,0,1>;
   surface := <4*t*cos(s), 2*t*sin(s), t*sin(s)>;
   trange := t=0..1;
   srange := s=0..2*Pi;
```

$$\begin{aligned} \text{vfield} &:= \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix} \\ \text{surface} &:= \begin{bmatrix} 4t \cos(s) \\ 2t \sin(s) \\ t \sin(s) \end{bmatrix} \\ \text{trange} &:= t = 0..1 \\ \text{srange} &:= s = 0..2\pi \end{aligned} \tag{1.1}$$

Evaluating the flux integral can be broken into a seven steps:


- 1) Graph the vector field and the parameterized surface to see what it looks like. (Not required, but useful.)
- 2) Set up the flux integral.
- 3) Replace x, y, and z in the vector field with the expressions in s and t on the surface.
- 4) Find the tangent vectors to the surface obtained by differentiating with respect to s and t.
- 5) Find a normal vector to the surface by taking the cross product of the tangent vectors.
- 6) Find the integrand of the flux integral by taking the dot product of the vector field and the normal vector.

7) Evaluate the integral.

>

The first step is to look at a graph of the vector field and the parameterized surface.

```
> vfieldplot := VectorField(vfield,output=plot,view=[-5..5,-5..5,-5..5]);
paramsurf := plot3d(surface,trange, srange,color=blue);
plots[display]({vfieldplot, paramsurf}, axes=boxed);
```



Our second step is to set up the flux integral.

```
> Int(Int((vfield.[Diff(surface,t) *X* Diff(surface,s)]),
trange), srange);
```

$$\int_0^{2\pi} \int_0^1 \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix} \cdot \left(\frac{\partial}{\partial t} \begin{bmatrix} 4t \cos(s) \\ 2t \sin(s) \\ t \sin(s) \end{bmatrix} \right) \times \left(\frac{\partial}{\partial s} \begin{bmatrix} 4t \cos(s) \\ 2t \sin(s) \\ t \sin(s) \end{bmatrix} \right) dt ds \quad (1.2)$$

The third step is to replace x, y, and z in the vector field with the parameterizations of x and y at the appropriate point on the surface.

```
> paramfield := evalVF(VectorField(vfield),surface);
print(`the function on the surface is `, paramfield);
```

$$\text{paramfield} := \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}$$

$$\text{the function on the surface is, } \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix} \quad (1.3)$$

The fourth step is to take the derivative of the parameterized surface with respect to the parameters s and t . This gives us two tangent vectors to the surface.

```
> tanvect := diff(surface, t):
tanvecs := diff(surface, s):
print(`the t-tangent vector to the surface is `, tanvect);
print(`the s-tangent vector to the surface is `, tanvecs);
```

$$\text{the } t\text{-tangent vector to the surface is, } \begin{bmatrix} 4 \cos(s) \\ 2 \sin(s) \\ \sin(s) \end{bmatrix}$$

$$\text{the } s\text{-tangent vector to the surface is, } \begin{bmatrix} -4 t \sin(s) \\ 2 t \cos(s) \\ t \cos(s) \end{bmatrix} \quad (1.4)$$

The fifth step is to take the cross product of those tangent vectors to get a normal vector.

```
> normvector := (tanvect &x tanvecs);
```

$$\text{normvector} := \begin{bmatrix} 0 \\ -4 \sin(s)^2 t - 4 \cos(s)^2 t \\ 8 \cos(s)^2 t + 8 \sin(s)^2 t \end{bmatrix} \quad (1.5)$$

The sixth step is to evaluate the dot product in the integrand and simplify. At this point we have reduced the flux integral over a surface to an ordinary iterated integral over a section of a plane.

```
> integrand := simplify(paramfield . normvector):
print(`The integrand is `, integrand);
```

$$\text{The integrand is, } 8 t \quad (1.6)$$

Finally we can evaluate the integral. We have Maple evaluate the integral in steps.

```
> Int(Int(integrand, trange), srange);
Int(int(integrand, trange), srange);
int(int(integrand, trange), srange);
```

$$\int_0^{2\pi} \int_0^1 8 t \, dt \, ds$$

$$\int_0^{2\pi} 4 \, ds$$

$$8 \pi$$

$$(1.7)$$

Exercise:

1) Let F be the field $[x, y, z]$. Let S be the surface $z = x^2 + y^2$ with $z \leq 4$. (You probably want to

parameterize the surface in terms of r and θ .) Compute the flux integral of F through S by hand. Then use Maple to check your calculations at each step.

[>

An automated approach

For convenience we block the code for those seven steps into two procedures we can use, one for plotting, and one for setting up the integral and evaluating. We also set up a black box procedure that evaluates without showing all the steps.

```
> surfaceplot := proc(vecfield, path, trange, srange,
                      viewwindow)
    local vfieldplot, paramsurf;
    vfieldplot := VectorField(vecfield,output=plot,
view=viewwindow):
    paramsurf := plot3d(surface,trange, srange, color=blue):
    plots[display]({vfieldplot, paramsurf}, axes=boxed);
end:
fluxintegral:= proc(vecfield, surface, trange, srange)
    local intval, paramfield, tanvect, tanvecs, normvector,
integrand;
    print(`the vector field `, vecfield);
    print(`the surface `, surface, ` with `, trange, ` and `,
srange);
    print(Int(Int((vfield.[Diff(surface,t)*X*Diff(surface,s)]
,
trange), srange));
    paramfield := evalVF(VectorField(vecfield),surface):
    print(`the vector field on the surface is `, paramfield);
    tanvect := diff(surface, t):
    tanvecs := diff(surface, s):
    print(`the t-tangent vector to the surface is `, tanvect);
    print(`the s-tangent vector to the surface is `, tanvecs);
    normvector := (tanvect &x tanvecs):
    print(`the normal vector to the surface is `, normvector);
    integrand := simplify(paramfield . normvector):
    print(`The integrand is `, integrand);
    print(Int(Int(integrand, trange), srange));
    print(Int(int(integrand, trange), srange));
    intval := int(int(integrand, trange), srange);
    print(`the integral is `, intval);
end:
```

With these procedures defined we can define a vector field and a parameterized surface and find the flux integral.

```
> vfield := <x, y, z>;
surface := <s*cos(t), s*sin(t), s^2>;
trange := t=0..2*Pi;
srange := s=0..2;
```

$$vfield := \begin{bmatrix} x \\ y \\ z \end{bmatrix}$$


$$surface := \begin{bmatrix} s \cos(t) \\ s \sin(t) \\ s^2 \end{bmatrix}$$

(3.1)

```

trange := t = 0 .. 2 * pi
srange := s = 0 .. 2
> surfaceplot(vfield, surface, trange, srange,
[-3..3, -3..3, -1..5]);
fluxintegral(vfield, surface, trange, srange);

```



the vector field, $\begin{bmatrix} x \\ y \\ z \end{bmatrix}$

the surface, $\begin{bmatrix} s \cos(t) \\ s \sin(t) \\ s^2 \end{bmatrix}$, with, $t = 0 .. 2 \pi$, and, $s = 0 .. 2$

$$\int_0^2 \int_0^{2\pi} \begin{bmatrix} x \\ y \\ z \end{bmatrix} \cdot \left[\left(\frac{\partial}{\partial t} \begin{bmatrix} s \cos(t) \\ s \sin(t) \\ s^2 \end{bmatrix} \right) \times \left(\frac{\partial}{\partial s} \begin{bmatrix} s \cos(t) \\ s \sin(t) \\ s^2 \end{bmatrix} \right) \right] dt ds$$

$$\begin{aligned} \text{the vector field on the surface is, } & \begin{bmatrix} s \cos(t) \\ s \sin(t) \\ s^2 \end{bmatrix} \\ \text{the } t\text{-tangent vector to the surface is, } & \begin{bmatrix} -s \sin(t) \\ s \cos(t) \\ 0 \end{bmatrix} \\ \text{the } s\text{-tangent vector to the surface is, } & \begin{bmatrix} \cos(t) \\ \sin(t) \\ 2s \end{bmatrix} \\ \text{the normal vector to the surface is, } & \begin{bmatrix} 2s^2 \cos(t) \\ 2s^2 \sin(t) \\ -s \sin(t)^2 - s \cos(t)^2 \end{bmatrix} \end{aligned}$$

The integrand is, s^3

$$\int_0^2 \int_0^{2\pi} s^3 dt ds$$

$$\int_0^2 2s^3 \pi ds$$

the integral is, 8π

(3.2)

Exercises

2) Let $F = [y, x, 0]$ and $S(s,t) = [3\sin(t), 3\cos(t), s+1]$ with $0 \leq t \leq 2\pi$ and $0 \leq s \leq 1$. Find the flux of the vector field through the surface.

[>

3) Let $F = [z, y, 2x]$ and S is the cone $z = \sqrt{x^2 + y^2}$ with $z \leq 2$. Find the flux of the vector field through the surface.

[>

4) The command `linalg[curl](F, [x, y, z])`; computes the curl of the vector F with respect to x , y , and z .

Let S be the sphere of radius 2. Describe S as a parameterized surface. (Think spherical.)

[>

Let $F = [xyz, x+y+z, x^3y^2z]$ and let G be the curl of F .

Compute the flux of G through S

[>

Explain why this nice result is not surprising.

[>

Just the result please

It is also useful to have a procedure that simply evaluates the flux integral. For that we use the `Flux` command from the `Student[VectorCalculus]` package.

```
> vfield := <x, y, z>;
```

```

surface := <s*cos(t), s*sin(t), s^2>;
trange := t=0..2*Pi;
srange := s=0..2;

```

$$vfield := \begin{bmatrix} x \\ y \\ z \end{bmatrix}$$

$$surface := \begin{bmatrix} s \cos(t) \\ s \sin(t) \\ s^2 \end{bmatrix}$$

$$trange := t = 0 .. 2 \pi$$

$$srange := s = 0 .. 2$$

(5.1)

```

> Flux(VectorField(vfield), Surface(surface, trange, srange),
output=integral);
Flux(VectorField(vfield), Surface(surface, trange, srange));

```

$$\int_0^{2\pi} \int_0^2 (2s^3 \cos(t)^2 + 2s^3 \sin(t)^2 + s^2 (-s \sin(t)^2 - s \cos(t)^2)) ds dt$$

$$8 \pi$$

(5.2)

Exercise

Exercise 5) Repeat exercise 4 with a closed surface of your choice and a vector field that is the curl of a nontrivial vector field of your choice.